

Application of wxMaxima System in LP problem of compound feed mass minimization

Włodzimierz Wojas¹, Jan Krupa²

¹ *Warsaw University of Life Sciences (SGGW), Poland, wlozdzimierz_wojas@sggw.pl*

² *Warsaw University of Life Sciences (SGGW), Poland, jan_krupa@sggw.pl*

In this paper we present application of CAS in teaching students of Warsaw University of Life Science. We apply wxMaxima system in the problem of fodders choice balancing minerals in compound feed of minimal total mass. We present procedure written in Maxima 5.26.0 which determines minimal quantities of fodders for supplying requirements of minerals for a selected type of animal.

1 Introduction

Ability to apply computational and programming tools offered by CAS systems, appears to be an important part of education in college of engineering. Students of engineering meet in the framework of core subjects a variety of practical problems: computational, optimization, data analysis and presentation, where the application of CAS systems can be very useful. Systems such as Mathematica, Maple, wxMaxima, Derive and others offer a whole range of different tools that can be used in practical engineering problems. This paper presents the application of the system wxMaxima in the problem of minimizing the mass of balanced compound feed. This problem was mandatory project for students of Agricultural and Forest Engineering at Warsaw University of Life Science within the subject Higher Mathematics II.

2 Formulation of optimization problem

The compound feed is composed of four fodders for one of the three types of animals in order to ensure daily requirements of four minerals (calcium, phosphorus, magnesium and sodium). The daily requirements of minerals for each of the three types of animals are presented in Table 1. Eleven types of fodders are presented in Table 2 together with the amounts in grams of minerals in 1kg of fodder dry matter. Students can choose four types of fodders which are combined into compound feed. After the choice of the type of animal and fodders the problem is to determine such quantities of four fodders that obtained compound feed contains acceptable

amounts of minerals (for daily requirements for one head of selected type of animal) and minimize total mass of compound feed. From mathematical point of view it is a LP problem with variables representing amounts of 1kg portions of fodders combined into compound feed. But the difficulty of the task lies not in solving the LP problem. The main difficulty is to choose the fodders of the compound feed for the selected type of animal that the LP problem has feasible solution.

Table 1: Animals daily minerals requirements

Type of animals	daily minerals requirements per head in grams			
	calcium	phosphorus	magnesium	sodium
Cattle	35	15–20	12–17	4.9
Sows	13	8	1–25	3.3–3.5
Sheep	2–4	1.5–2.5	1–3	10

Table 2: Amounts of minerals (g) in 1 kg of fodder

Index	Fodder	In 1 kg dry matter			
		calcium	phosphorus	magnesium	sodium
1	corn	0.18	3.0	1.36	0.4
2	potatoes	0.57	1.83	1.13	0.6
3	wheat bran	1.32	13.50	5.1	1.2
4	oat	1.37	4.2	1.35	1.3
5	straw	2.43	0.79	0.80	2.8
6	linseed oil cake	3.43	11.14	5.59	1.0
7	meat meal	6.79	25.3	2.9	11.9
8	meadow hay	7.15	2.53	2.91	0.8
9	sugar beet leaves	12.40	3.1	3.9	5.5
10	alfalfa hay	18.30	2.60	2.19	0.6
11	low fat milk	12.43	9.52	1.2	6.2

Example 1. To illustrate the problem, assume that selected type of animal is sheep. Assume also that we choose: corn, potatoes, wheat bran and oat to combine them into compound feed. Let x_1, x_2, x_3, x_4 are amounts of 1kg portions of corresponding fodders i.e. corn, potatoes, wheat bran and oat. To guarantee acceptable amounts of minerals for daily requirements and minimize total mass of compound feed, we have to solve the following LP problem:

$$\begin{array}{ll}
\text{Minimize} & x_1 + x_2 + x_3 + x_4 \\
\text{Subject to} & 0.18x_1 + 0.57x_2 + 1.32x_3 + 1.37x_4 \geq 2 \\
& 0.18x_1 + 0.57x_2 + 1.32x_3 + 1.37x_4 \leq 4 \\
& 3x_1 + 1.83x_2 + 13.5x_3 + 4.2x_4 \leq 2.5 \\
& 3x_1 + 1.83x_2 + 13.5x_3 + 4.2x_4 \geq 1.5 \\
& 1.36x_1 + 1.13x_2 + 5.1x_3 + 1.35x_4 \geq 1 \\
& 1.36x_1 + 1.13x_2 + 5.1x_3 + 1.35x_4 \leq 3 \\
& 0.4x_1 + 0.6x_2 + 1.2x_3 + 1.3x_4 = 10 \\
& x_i \geq 0 \text{ for } i = 1, 2, 3, 4.
\end{array}$$

Applying the standard function `minimize_lp(obj, cond,[pos])` in Maxima 5.26.0 we get: no feasible solution. The answer is negative - there is no possibility to balance minerals in compound feed for these four fodders and daily requirements of minerals for sheep.

3 Numerical procedure to solve the problem

To determine feasible solution for a selected type of animal, students usually try from few to several times considering different types of fodders. This problem can be solved quite easily using optimization procedure `minimal_mass(A,b1,b2,m)` which we wrote in Maxima 5.26.0 programming language. The simpler version of this problem is presented and solved using Mathematica 8.0 in paper [11]. The principles of programming in Maxima 5.26.0 are presented in [2, 3, 4, 5, 7, 8]. The procedure `minimal_mass(A,b1,b2,m)` is a function returning as a last component vector of pairs which first element is a number of fodder (from Table 2) and the second is the amount of this fodder combined into compound feed. If the problem has no feasible solution the function returns statement: "no feasible solution!" The amounts of fodders which we obtained have minimal total sum. A is a matrix of amounts of minerals in fodders, created on base of Table 2. $b1$, $b2$ are vectors of minimal and maximal respectively, daily requirement of minerals for selected type of animal created on base of Table 1. m is a number of fodders in compound feed. In our problem $m = 4$. Formal description of the procedure `minimal_mass(A,b1,b2,m)` is presented below.

Listing 1: Description of the procedure `minimal_mass`:

```
minimal_mass(A, b1, b2, m) := block ([ n, obj, p, i, L, R1, R2, R, xi,
```

```

Ai,A1,A2,T,x], /* local variables */
/* you have to load the simplex package
load( "simplex" ); */
n:length(A), /* n=numbers of rows in marix A */
xi: makelist( concat(x,i),i,1,m), /* xi=[x1,x2,...,xm] */
obj: sum(xi[i],i,1,m), /* obj= x1+x2+...+xm */
L: makelist(i,i,1,n), /* L=[1,2,...,xn], L is
a Maxima list */
L: setify(L), /* L={1,2,...,n}, L is Maxima set,
in order to use the function powerset */
L: powerset(L,m), /* L= all m subsets of set L */
L: full_listify(L), /* set L is converted to
a Maxima list L */
i:0, /* initial number of iterations is 0 */
T:false, /* the loop below will run at least once */
for Ai in L while is(T=false) do (
A1: makelist(A[x],x,Ai), /* Ai is a list of m positive
integers [n_{i1},...,n_{im}], A1= m x m matrix */
R1: makelist(sum(A1[i][j]*xi[i],i,1,m)>=b1[j],j,1,m),
/* above: we construct the list of constrain
inequalities: a_{1j}*x_1+...+a_{mj}*x_m>=b1[j], j=1,2,...,m */
R2: makelist(sum(A1[i][j]*xi[i],i,1,m)<=b2[j],j,1,m),
/* above: we construct the list of constrain
inequalities: a_{1j}*x_1+...+a_{mj}*x_m<=b2[j],
j=1,2,...,m */
R: flatten([R1,R2]), /* R is a list of all constrains */
p: minimize_lp(obj,R,xi), /* we call the Maxima
function minimize_lp, the third argument xi means
that all variables in list xi are nonnegative,
p=[minimal value of object function,
[xm=value_m,...,x1=value_1] */
if (is(p#"Problem not feasible!") and is(p#"Problem
not bounded!")) then
(T:true,
p1:p, /* we save p in case we obtain minimal
feasible solution p */
p: makelist(rhs(p[2][m-i]),i,0,m-1),
/* p[2]=[xm=value_m,...,x1=value_1], so we obtain
p=[value_1,...,value_m] */
A2:Ai /* we save Ai in case we obtain minimal

```

```

feasible solution p */
),
i:i+1 /* number of iteration increased */
),
(if is(T=true) then
/* if T=true we obtain minimal feasible solution p */
[i, sol=p1, transpose(p), map(lambda([x,y],[x,y]),A2,p)]
else [i,"no nonnegative solution",p])
)$;

```

Example 2. Assume that selected type of animal is cattle. We define A , b_1 , b_2 , m in Figure 1.

$A : [[0.18, 3.0, 1.36, 0.4], [0.57, 1.83, 1.13, 0.6], [1.32, 13.50, 5.1, 1.2], [1.37, 4.2, 1.35, 1.3], [2.43, 0.79, 0.80, 2.8], [3.43, 11.14, 5.59, 1.0], [6.79, 25.3, 2.9, 11.9], [7.15, 2.53, 2.91, 0.8], [12.40, 3.1, 3.9, 5.5], [18.30, 2.60, 2.19, 0.6], [12.43, 9.52, 1.2, 6.2]]; b_1 : [35, 15, 12, 4.9]; b_2:[35,20,17,4.9]; m:4;$

Figure 1: Matrix A , vectors b_1 , b_2 and m .

The computations were conducted in Maxima5.26.0 on PC with 2.6 GHz, Pentium 4 processor and 512 Mb RAM under Windows XP platform. Maxima elapsed_run_time was 0.07 s. The final result is presented in Figure 2.

$[5, sol = [6.191975412126295, [x_4 = 4.737686504610226, x_3 = 0.3954624196703, x_2 = 1.058826487845771, x_1 = 0]],$ $\begin{pmatrix} 0 \\ 1.058826487845771 \\ 0.3954624196703 \\ 4.737686504610226 \end{pmatrix},$ $[[1, 0], [2, 1.058826487845771], [3, 0.3954624196703], [8, 4.737686504610226]]]$
--

Figure 2: Final solution of the problem.

This solution was obtained in fifth iteration. First four iterations had no feasible solutions. The obtained result gives us the following amounts of fodders (see column Index in Table 2) : $[1, 0]$ means: 0.0000 kg of meadow, $[2, 1.0588]$ means 1.0588 kg of wheat bran, $[3, 0.3955]$ means 0.3955 kg of potatoes and $[8, 4.7377]$ means 4.7377 kg of corn. Total mass of compound feed is 6.192 kg. It is minimal

mass of compound feed comprised of these four fodders which guarantees daily requirement of four minerals for one cattle.

The procedure `minimal_mass(A,b1,b2,m)` can be easily extended to minimize total mass of compound feed among all possible compound feeds balancing minerals for selected type of animal. The extended procedure is in Figure Listing 2

Listing 2: Description of the procedure `minimal_mass1`:

```
minimal_mass1(A,b1,b2,m):= block ([ n , obj , p , p1 , p0 , i , i0 , L ,
R1 , R2 , R , xi , Ai , A1 , A2 , T , x , y0 ] ,
n : length (A) ,
xi : makelist (concat (x , i) , i , 1 , m) ,
obj : sum (xi [ i ] , i , 1 , m) ,
L : makelist (i , i , 1 , n) ,
L : setify (L) ,
L : powerset (L , m) ,
L : full_listify (L) ,
i : 0 ,
i0 : 0 ,
y0 : 100 ,
T : false ,
for Ai in L do (
/* in minimal_mass we have: for Ai in L while is(T=false):
now the condition "while is(T=false)" is not needed */
A1 : makelist (A[x] , x , Ai) ,
R1 : makelist (sum (A1 [ i ] [ j ] * xi [ i ] , i , 1 , m) >= b1 [ j ] , j , 1 , m) ,
R2 : makelist (sum (A1 [ i ] [ j ] * xi [ i ] , i , 1 , m) <= b2 [ j ] , j , 1 , m) ,
R : flatten ([ R1 , R2 ] ) ,
p : minimize_lp (obj , R , xi) ,
if (is (p#"Problem not feasible!") and is (p#"Problem
not bounded!")) then
(T : true ,
if is (p[1] < y0) then (
y0 : p[1] ,
p1 : p ,
p0 : makelist (rhs (p[2][ i ] ) , i , 1 , m) ,
A2 : Ai ,
i0 : i
)) ,
i : i+1
```

```

),
(if is(T=true) then
[i, i0, sol=p1, transpose(p0), map(lambda ([x,y],[x,y]),A2,p0)]
else [i,"no nonnegative solution",p0])
);

```

The computations were conducted in Maxima5.26.0 on PC with 2.6 GHz, Pentium 4 processor and 512 Mb RAM under Windows XP platform. Maxima elapsed_run_time was 1.83 s. The final result is presented in Figure 3.

Applying the procedure minimal_mass1(A,b1,b2,m) to the extension of the problem for cattle we obtain the solution presented in Figure 3.

$$\begin{aligned}
& [330, 321, sol = [3.108826328073963, [x_4 = 1.325453353602, \\
& \quad x_3 = 0.51585666963707, x_2 = 0, x_1 = 1.267516304834889]], \\
& \quad \begin{pmatrix} 1.325453353602 \\ 0.51585666963707 \\ 0 \\ 1.267516304834889 \end{pmatrix} \\
& [[6, 1.325453353602], [8, 0.51585666963707], [9, 0], [10, 1.267516304834889]]]
\end{aligned}$$

Figure 3: Final solution of the generalized problem with cattle.

This solution was obtained in 321-th iteration. First 320 iterations give no feasible solutions or feasible solutions with value of object function greater than 3.1088. The obtained result gives us the following amounts of fodders (see column Index in Table 2): [6, 1.32545] means: 1.32545 kg of linseed oil cake, [8, 0.515856669637] means 0.515856669637 kg of meadow hay, [9, 0] means 0.0000 kg of sugar beet leaves and [10, 1.2675163] means 1.2675163 kg of alfalfa hay. Total mass of compound feed is 3.1088 kg. It is minimal mass of compound feed comprised of these four fodders chosen from all eleven fodders, which guarantees daily requirement of four minerals for one cattle.

4 Summary

In this paper the procedure minimal_mass(A,b1,b2,m) implemented in Maxima 5.26.0 was presented. This procedure solves the problem of fodders choice for minimal mass compound feed in order to ensure daily requirement of four minerals (calcium, phosphorus, magnesium and sodium) for a selected type of animal. This problem was assignment for students Agricultural and Forest Engineering at

Warsaw University of Life Science. Finding solution of the problem students need quite much of work considering various selection of four fodders in order to obtain balancing minerals in compound feed. Applying CAS system such as wxMaxima allows find solution of described above problem in a much less time.

References

- [1] Homepage of the Maxima Project: <http://maxima.sourceforge.net>
- [2] *Maxima Documentation* at <http://maxima.sourceforge.net/documentation.html>.
- [3] *Maxima 5.36.0 Manual* at <http://maxima.sourceforge.net/docs/manual/maxima.html>.
- [4] Edwin L. Woollett, *Maxima by Example* at <http://web.csulb.edu/~woollett/#mbe>
- [5] Rafał Topolnicki, *Maxima - practical guide* (in polish) at <http://www.knf.ifd.uni.wroc.pl/materialy/maxima.pdf>
- [6] Robert Dodier, *Minimal Maxima* at <http://maxima.sourceforge.net/docs/tutorial/en/minimal-maxima.pdf>
- [7] Paulo Ney de Souza, Richard J. Fateman, Joel Moses, Cliff Yapp, *The Maxima Book* at <http://maxima.sourceforge.net/docs/maximabook/maximabook-19-Sept-2004.pdf>
- [8] Gilberto E. Urroz, *MAXIMA: Science and Engineering Applications* at <http://www.neng.usu.edu/cee/faculty/gurro/Maxima.html>
- [9] Jaime E. Villate, *Dynamical systems* (in Portuguese) at http://www.villate.org/doc/sistemasdinamicos/sistdinam-1_2.pdf
- [10] Homepage of the wxMaxima Project: <http://andrejv.github.io/wxmaxima/>
- [11] Włodzimierz Wojas, Jan Krupa, *Application of Mathematica System in the problem of balance of ingredients for compound feed*, Computer Algebra Systems In Teaching and Research 2013, Vol. 4, No 1, pp. 278-282
- [12] Ziiolecka A., Kuźdowicz M., Chomszyn M. *Tables of ingredients of domestic fodders* (in polish) PWN Warszawa (1987)
- [13] The minerals in animal nutrition. Poznan University of Life Science (1994) (in polish)
- [14] www.dodatkipaszowe.pl